The Stochastic Power Law Logistic Model: Derivations of ODEs for the first 3 Cumulants.

Ingemar Nåsell

Three Maple procedures are given in this Maple worksheet. We give first 2 Maple procedures that allow us to express cumulants in terms of (raw) moments, and vice versa. The procedure **cumf** determines the cumulant of order kk as a function of the (raw) moments of orders up to kk. Similarly, the procedure **muf** determines the (raw) moment of order kk as a function of the cumulants of orders up to kk. In both of these procedures, M denotes the moment generating function, and K denotes the cumulant generating function. Furthermore, Ks is the sum of the first kk terms of the series expansion of K, and Ms is the sum of the first kk+1 terms of the series expansion of M.

After the presentation of the two proedures **cumf** and **muf** follows a sequence of commands that together derive the ODEs of the first 3 cumulants. It turns out that the result of a sequence of 5 commands that are executed sequentially can be effected by a single Maple command. We use this finding to establish a Maple procedure **ODEcum3** that derives the ODEs of the first 3 cumulants.

The main results derived in the manuscript that we deal with here is asymptotic approximations of the first 3 cumulants. The derivations are given by 10 Maple worksheets, one for each of the integer values 1,2,...,10 of s. Each of these worksheets contains the 3 procedures given here. Clearly, derivations of ODEs of the first 3 cumulants is a necessary first step in the derivations of our results. The main purpose of the present worksheet is to give insight into the arguments that are used to establish the procedure **ODEcum3.** Some of the commands in this procedure are rather compact and not intuitively appealing. The long derivations of these commands will not be repeated in the 10 worksheets that do the main job of deriving asymptotic approximations of the first 3 cumulants.

```
> restart;
```

```
> cumf:=proc(kk)
   local M,K,Ks,cum;
   description "Determines the cumulant of order kk as function of
  the raw moments of orders up to kk";
   M:=1+add(cat(mu,k)*theta^k/k!,k=1..kk);
   K:=log(M);
   Ks:=convert(series(K,theta,kk+1),polynom);
   cum:=sort(simplify(coeff(Ks,theta,kk)*kk!),[seq(cat(mu,j),j=1..
  kk)],plex);
  end proc:
```

```
> muf:=proc(kk)
```

```
  local K,M,Ms,mu;
  description "Determines the (raw) moment of order kk as function
of the cumulants of orders up to kk";
  K:=add(cat(kappa,k)*theta^k/k!,k=1..kk);
  M:=exp(K);
  Ms:=convert(series(M,theta,kk+1),polynom);
  mu:=sort(coeff(Ms,theta,kk)*kk!,[seq(cat(kappa,j),j=1..kk)],
plex);
end proc:
```

The goal of the third procedure is to derive ODEs for the first 3 cumulants. They take the form of time derivatives of the first 3 cumulants, as functions of the cumulants. The starting points are expressions for the first 3 cumulants as functions of the (raw) moments. They are determined by the procedure **cumf**. The time derivatives of the cumulants are then determined. They are expressed with terms that contain both moments and derivatives of moments. The derivatives of the moments are expressed in terms of moments by using results given by Matis-Kiffe-Parthasarathy (1998). After this operation, the expressions for the 3 time derivatives of the cumulants are given in terms of moments. The last step is to express the moments in terms of cumulants, using the procedure **muf**.

Expressions for the first 3 cumulants in terms of the moments are:

```
> cumf(1);
  cumf(2);
  cumf(3);
```

$$\mu 1$$

$$-\mu 1^2 + \mu 2$$

$$2\,\mu 1^3 - 3\,\mu 1\,\mu 2 + \mu 3 \tag{1}$$

The derivatives of these expressions are denoted Dcum1, Dcum2, Dcum3. Different versions of these derivatives are denoted Dcum1a, Dcum1b, Dcum1c, etc:

```
> Dcum1a:=D(cumf(1));
  Dcum2a:=D(cumf(2));
  Dcum3a:=D(cumf(3));
```

$$Dcum1a := \mathrm{D}(\mu 1)$$

$$Dcum2a := -2\,\mathrm{D}(\mu 1)\,\mu 1 + \mathrm{D}(\mu 2)$$

$$Dcum3a := 6\,\mathrm{D}(\mu 1)\,\mu 1^2 - 3\,\mathrm{D}(\mu 1)\,\mu 2 - 3\,\mu 1\,\mathrm{D}(\mu 2) + \mathrm{D}(\mu 3) \tag{2}$$

Expressions for the derivatives of the moments are given by Matis-Kiffe-Parthasarathy (1998):

```
> Dmu1:=a*mu1-b*cat(mu,s+1);
  Dmu2:=2*a*mu2+c*mu1-d*cat(mu,s+1)-2*b*cat(mu,s+2);
  Dmu3:=a*(mu1+3*mu3)+3*c*mu2-3*d*cat(mu,s+2)-b*(cat(mu,s+1)+3*cat
  (mu,s+3));
```

$$Dmu1 := a\,\mu 1 - b\,mus{+}1$$

$$Dmu2 := 2\,a\,\mu 2 - 2\,b\,mus{+}2 + c\,\mu 1 - d\,mus{+}1$$

$$Dmu3 := a\,(\mu 1 + 3\,\mu 3) + 3\,c\,\mu 2 - 3\,d\,mus{+}2 - b\,(mus{+}1 + 3\,mus{+}3) \tag{3}$$

Next, substitute these expressions for the moment derivatives into the expressions for the cumulant derivatives. The 3 substitutions are written as follows:

```
> sub1:=D(mu1)=Dmu1,D(mu2)=Dmu2,D(mu3)=Dmu3;
```

$$sub1 := \mathrm{D}(\mu 1) = a\,\mu 1 - b\,mus{+}1,\ \mathrm{D}(\mu 2) = 2\,a\,\mu 2 - 2\,b\,mus{+}2 + c\,\mu 1 - d\,mus{+}1,\ \mathrm{D}(\mu 3) = a\,(\mu 1 + 3\,\mu 3) + 3\,c\,\mu 2 - 3\,d\,mus{+}2 - b\,(mus{+}1 + 3\,mus{+}3) \tag{4}$$

Versions of the cumulant derivatives are found by using the Maple command subs:

```
> Dcum1b:=subs(sub1,Dcum1a);
  Dcum2b:=subs(sub1,Dcum2a);
  Dcum3b:=subs(sub1,Dcum3a);
```

$$Dcum1b := a\,\mu 1 - b\,mus{+}1$$

$$Dcum2b := -2\,(a\,\mu 1 - b\,mus{+}1)\,\mu 1 + 2\,a\,\mu 2 - 2\,b\,mus{+}2 + c\,\mu 1 - d\,mus{+}1$$

$$Dcum3b := 6\,(a\,\mu 1 - b\,mus{+}1)\,\mu 1^2 - 3\,(a\,\mu 1 - b\,mus{+}1)\,\mu 2 - 3\,\mu 1\,(2\,a\,\mu 2 - 2\,b\,mus{+}2 + c\,\mu 1 - d\,mus{+}1) + a\,(\mu 1 + 3\,\mu 3) + 3\,c\,\mu 2 - 3\,d\,mus{+}2 - b\,(mus{+}1 + 3\,mus{+}3) \tag{5}$$

As a preparation for the next substitution command, we choose a particular s-value:

```
> s:=1;
```

$$s := 1 \tag{6}$$

The next substitution consists in expressing the moments in terms of cumulants.

```
> sub2:=seq(cat(mu,i)=muf(i),i=1..3+s);
```

$$sub2 := \mu 1 = \kappa 1,\ \mu 2 = \kappa 1^2 + \kappa 2,\ \mu 3 = \kappa 1^3 + 3\,\kappa 1\,\kappa 2 + \kappa 3,\ \mu 4 = \kappa 1^4 + 6\,\kappa 1^2\,\kappa 2 + 4\,\kappa 1\,\kappa 3 + 3\,\kappa 2^2 + \kappa 4 \tag{7}$$

By again using the Maple command subs, we get the next set of versions of the cumulant derivatives:

```
> Dcum1c:=subs(sub2,Dcum1b);
  Dcum2c:=subs(sub2,Dcum2b);
  Dcum3c:=subs(sub2,Dcum3b);
```

$$Dcum1c := a\,\kappa 1 - b\,mus{+}1$$

$$Dcum2c := -2\,(a\,\kappa 1 - b\,mus{+}1)\,\kappa 1 + 2\,a\,(\kappa 1^2 + \kappa 2) - 2\,b\,mus{+}2 + c\,\kappa 1 - d\,mus{+}1$$

$$Dcum3c := 6\,(a\,\kappa 1 - b\,mus{+}1)\,\kappa 1^2 - 3\,(a\,\kappa 1 - b\,mus{+}1)\,(\kappa 1^2 + \kappa 2) - 3\,\kappa 1\,(2\,a\,(\kappa 1^2 + \kappa 2) - 2\,b\,mus{+}2 + c\,\kappa 1 - d\,mus{+}1) + a\,(3\,\kappa 1^3 + 9\,\kappa 1\,\kappa 2 + \kappa 1 + 3\,\kappa 3) + 3\,c\,(\kappa 1^2 + \kappa 2) - 3\,d\,mus{+}2 - b\,(mus{+}1 + 3\,mus{+}3) \tag{8}$$

In each of the 3 cases, we collect in terms of the parameters a, c, d, b:

```
> Dcum1d:=collect(Dcum1c,[a,c,d,b]);
  Dcum2d:=collect(Dcum2c,[a,c,d,b]);
  Dcum3d:=collect(Dcum3c,[a,c,d,b]);
```

$$Dcum1d := a\,\kappa 1 - b\,mus{+}1$$

$$Dcum2d := 2\,\kappa 2\,a + c\,\kappa 1 - d\,mus{+}1 + (2\,\kappa 1\,mus{+}1 - 2\,mus{+}2)\,b$$

$$Dcum3d := \left(9\,\kappa1^3 - 3\,\kappa1\left(\kappa1^2+\kappa2\right) - 3\,\kappa1\left(2\,\kappa1^2+2\,\kappa2\right) + 9\,\kappa1\,\kappa2 + \kappa1 + 3\,\kappa3\right) a + 3\,\kappa2\,c + \left(3\,\kappa1\,mus{+}1 - 3\,mus{+}2\right) d + \left(-6\,mus{+}1\,\kappa1^2 + 3\,mus{+}1\left(\kappa1^2+\kappa2\right) + 6\,\kappa1\,mus{+}2 - mus{+}1 - 3\,mus{+}3\right) b \quad \textbf{(9)}$$

The factors that multiply each of the parameters a, c, d, b are simplified as follows:

```
> Dcum1e:=map(simplify,Dcum1d);
  Dcum2e:=map(simplify,Dcum2d);
  Dcum3e:=map(simplify,Dcum3d);
```

$$Dcum1e := a\,\kappa1 - b\,mus{+}1$$

$$Dcum2e := 2\,\kappa2\,a + c\,\kappa1 - d\,mus{+}1 + 2\left(\kappa1\,mus{+}1 - mus{+}2\right) b$$

$$Dcum3e := \left(\kappa1 + 3\,\kappa3\right) a + 3\,\kappa2\,c + 3\left(\kappa1\,mus{+}1 - mus{+}2\right) d - 3\left(\left(\kappa1^2 - \kappa2 + \frac{1}{3}\right) mus{+}1 - 2\,\kappa1\,mus{+}2 + mus{+}3\right) b \quad \textbf{(10)}$$

The factors that multiply the parameters a, c, d, b are polynomials in the cumulants. We sort them as follows:

```
> Dcum1f:=map(sort,Dcum1e);
  Dcum2f:=map(sort,Dcum2e);
  Dcum3f:=map(sort,Dcum3e);
```

$$Dcum1f := a\,\kappa1 - b\,mus{+}1$$

$$Dcum2f := 2\,a\,\kappa2 + c\,\kappa1 - d\,mus{+}1 + 2\left(\kappa1\,mus{+}1 - mus{+}2\right) b$$

$$Dcum3f := \left(\kappa1 + 3\,\kappa3\right) a + 3\,c\,\kappa2 + 3\left(\kappa1\,mus{+}1 - mus{+}2\right) d - 3\left(-2\,\kappa1\,mus{+}2 + \left(\kappa1^2 - \kappa2 + \frac{1}{3}\right) mus{+}1 + mus{+}3\right) b \quad \textbf{(11)}$$

We note now that the sequence of 5 operations that we have carried out above, namly the 2 sustitutions that use the command subs, the collection in terms of the parameters a, c, d, b, the simplifications of the factors of these parameters, and the sorting as polynomials can all be effected in one step in Maple, as follows:

```
> Dcum1:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
  (1)))),[a,c,d,b])));
  Dcum2:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
  (2)))),[a,c,d,b])));
  Dcum3:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
  (3)))),[a,c,d,b])));
```

$$Dcum1 := a\,\kappa1 - b\,mus{+}1$$

$$Dcum2 := 2\,a\,\kappa2 + c\,\kappa1 - d\,mus{+}1 + 2\left(\kappa1\,mus{+}1 - mus{+}2\right) b$$

$$Dcum3 := \left(\kappa1 + 3\,\kappa3\right) a + 3\,c\,\kappa2 + 3\left(\kappa1\,mus{+}1 - mus{+}2\right) d - 3\left(-2\,\kappa1\,mus{+}2 + \left(\kappa1^2 - \kappa2 + \frac{1}{3}\right) mus{+}1 + mus{+}3\right) b \quad \textbf{(12)}$$

We proceed to formulate a Maple procedure that determines the ODEs for the first 3 cumulants. We introduce the reparametrization that expresses the parameters a,b,c,d in terms of the parameters mu, R0, alpha, N. By using the above result, we find that this Maple procedure can be written as follows:

```
> ODEcum3:=proc(s)
    local Dmu1,Dmu2,Dmu3,sub1,sub2,Dcum1,Dcum2,Dcum3,a,c,d,b;
    description "Derive ODEs for the first 3 cumulants";
    Dmu1:=a*mu1-b*cat(mu,s+1);
    Dmu2:=2*a*mu2+c*mu1-d*cat(mu,s+1)-2*b*cat(mu,s+2);
    Dmu3:=a*(mu1+3*mu3)+3*c*mu2-3*d*cat(mu,s+2)-b*(cat(mu,s+1)+3*
  cat(mu,s+3));
    sub1:=D(mu1)=Dmu1,D(mu2)=Dmu2,D(mu3)=Dmu3;
    sub2:=seq(cat(mu,i)=muf(i),i=1..3+s);
    Dcum1:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
  (1)))),[a,c,d,b])));
    Dcum2:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
  (2)))),[a,c,d,b])));
    Dcum3:=map(sort,map(simplify,collect(subs(sub2,subs(sub1,D(cumf
  (3)))),[a,c,d,b])));
    a:=mu*(R0-1);
    c:=mu*(R0+1);
    d:=mu*(R0-alpha)/N^s;
    b:=mu*(R0+alpha)/N^s;
    [eval(Dcum1),eval(Dcum2),eval(Dcum3)];
  end proc:
```

We use the above procedure to determine the ODEs for the first 3 cumulants for the s-values 1, 2, 3, and 4:

```
> ODEcum3(1);
```

$$\left[\mu\,(R0-1)\,\kappa 1-\frac{(\kappa 1^2+\kappa 2)\,\mu\,(R0+\alpha)}{N},\ 2\,\mu\,(R0-1)\,\kappa 2+\mu\,(R0+1)\,\kappa 1-\frac{(\kappa 1^2+\kappa 2)\,\mu\,(R0-\alpha)}{N}-\frac{4\,\mu\,(R0+\alpha)\,\kappa 1\,\kappa 2}{N}-\frac{2\,\mu\,(R0+\alpha)\,\kappa 3}{N},\ (\kappa 1+3\,\kappa 3)\,\mu\,(R0-1)+3\,\mu\,(R0+1)\,\kappa 2-\frac{6\,\mu\,(R0-\alpha)\,\kappa 1\,\kappa 2}{N}-\frac{3\,\mu\,(R0-\alpha)\,\kappa 3}{N}-\frac{(\kappa 1^2+6\,\kappa 1\,\kappa 3+6\,\kappa 2^2+\kappa 2+3\,\kappa 4)\,\mu\,(R0+\alpha)}{N}\right] \quad \mathbf{(13)}$$

```
> ODEcum3(2);
```

$$\left[\mu\,(R0-1)\,\kappa 1-\frac{(\kappa 1^3+3\,\kappa 1\,\kappa 2+\kappa 3)\,\mu\,(R0+\alpha)}{N^2},\ 2\,\mu\,(R0-1)\,\kappa 2+\mu\,(R0+1)\,\kappa 1-\frac{(\kappa 1^3+3\,\kappa 1\,\kappa 2+\kappa 3)\,\mu\,(R0-\alpha)}{N^2}\right. \quad \mathbf{(14)}$$

$$- \frac{6 \left(\kappa 1^2 \kappa 2 + \kappa 1 \kappa 3 + \kappa 2^2 + \frac{1}{3} \kappa 4\right) \mu (R0 + \alpha)}{N^2}, (\kappa 1 + 3 \kappa 3) \mu (R0 - 1)$$

$$+ 3 \mu (R0 + 1) \kappa 2 - \frac{9 \left(\kappa 1^2 \kappa 2 + \kappa 1 \kappa 3 + \kappa 2^2 + \frac{1}{3} \kappa 4\right) \mu (R0 - \alpha)}{N^2}$$

$$- \frac{1}{N^2} \left(\left(\kappa 1^3 + 9 \kappa 1^2 \kappa 3 + 27 \kappa 2 \kappa 3 + \left(18 \kappa 2^2 + 3 \kappa 2 + 9 \kappa 4\right) \kappa 1 + \kappa 3 \right.\right.$$

$$\left.\left. + 3 \kappa 5\right) \mu (R0 + \alpha) \right) \Big]$$

```
> ODEcum3(3);
```

$$\Big[\mu (R0 - 1) \kappa 1 - \frac{\left(\kappa 1^4 + 6 \kappa 1^2 \kappa 2 + 4 \kappa 1 \kappa 3 + 3 \kappa 2^2 + \kappa 4\right) \mu (R0 + \alpha)}{N^3}, 2 \mu (R0 \qquad \textbf{(15)}$$

$$- 1) \kappa 2 + \mu (R0 + 1) \kappa 1 - \frac{\left(\kappa 1^4 + 6 \kappa 1^2 \kappa 2 + 4 \kappa 1 \kappa 3 + 3 \kappa 2^2 + \kappa 4\right) \mu (R0 - \alpha)}{N^3}$$

$$- \frac{2 \left(4 \kappa 1^3 \kappa 2 + 6 \kappa 1^2 \kappa 3 + 12 \kappa 1 \kappa 2^2 + 4 \kappa 1 \kappa 4 + 10 \kappa 2 \kappa 3 + \kappa 5\right) \mu (R0 + \alpha)}{N^3},$$

$$(\kappa 1 + 3 \kappa 3) \mu (R0 - 1) + 3 \mu (R0 + 1) \kappa 2$$

$$- \frac{1}{N^3} \left(3 \left(4 \kappa 1^3 \kappa 2 + 6 \kappa 1^2 \kappa 3 + 12 \kappa 1 \kappa 2^2 + 4 \kappa 1 \kappa 4 + 10 \kappa 2 \kappa 3 \right.\right.$$

$$\left.\left. + \kappa 5\right) \mu (R0 - \alpha) \right) - \frac{1}{N^3} \left(\left(\kappa 1^4 + 12 \kappa 1^3 \kappa 3 + 36 \kappa 2^3 + \left(36 \kappa 2^2 + 6 \kappa 2 \right.\right.\right.$$

$$\left. + 18 \kappa 4\right) \kappa 1^2 + 3 \kappa 2^2 + 42 \kappa 2 \kappa 4 + 30 \kappa 3^2 + (108 \kappa 2 \kappa 3 + 4 \kappa 3 + 12 \kappa 5) \kappa 1 + \kappa 4$$

$$\left.\left. + 3 \kappa 6\right) \mu (R0 + \alpha) \right) \Big]$$

```
> ODEcum3(4);
```

$$\Big[\mu (R0 - 1) \kappa 1 \qquad \textbf{(16)}$$

$$- \frac{1}{N^4} \left(\left(\kappa 1^5 + 10 \kappa 1^3 \kappa 2 + 10 \kappa 1^2 \kappa 3 + 10 \kappa 2 \kappa 3 + \left(15 \kappa 2^2 + 5 \kappa 4\right) \kappa 1 \right.\right.$$

$$\left.\left. + \kappa 5\right) \mu (R0 + \alpha) \right), 2 \mu (R0 - 1) \kappa 2 + \mu (R0 + 1) \kappa 1$$

$$
- \frac{1}{N^4} \left(\left(\kappa 1^5 + 10\, \kappa 1^3\, \kappa 2 + 10\, \kappa 1^2\, \kappa 3 + 10\, \kappa 2\, \kappa 3 + \left(15\, \kappa 2^2 + 5\, \kappa 4\right) \kappa 1
+ \kappa 5\right) \mu\left(R0 - \alpha\right) \right) - \frac{1}{N^4} \left(10 \left(\kappa 1^4\, \kappa 2 + 2\, \kappa 1^3\, \kappa 3 + 3\, \kappa 2^3 + \left(6\, \kappa 2^2
+ 2\, \kappa 4\right) \kappa 1^2 + 3\, \kappa 2\, \kappa 4 + 2\, \kappa 3^2 + \left(10\, \kappa 2\, \kappa 3 + \kappa 5\right) \kappa 1 + \frac{\kappa 6}{5} \right) \mu\left(R0 + \alpha\right) \right), \left(\kappa 1
+ 3\, \kappa 3\right) \mu\left(R0 - 1\right) + 3\, \mu\left(R0 + 1\right) \kappa 2 - \frac{1}{N^4} \left(15 \left(\kappa 1^4\, \kappa 2 + 2\, \kappa 1^3\, \kappa 3 + 3\, \kappa 2^3
+ \left(6\, \kappa 2^2 + 2\, \kappa 4\right) \kappa 1^2 + 3\, \kappa 2\, \kappa 4 + 2\, \kappa 3^2 + \left(10\, \kappa 2\, \kappa 3 + \kappa 5\right) \kappa 1 + \frac{\kappa 6}{5} \right) \mu\left(R0
- \alpha\right) \right) - \frac{1}{N^4} \left(\left(\kappa 1^5 + 15\, \kappa 1^4\, \kappa 3 + \left(60\, \kappa 2^2 + 10\, \kappa 2 + 30\, \kappa 4\right) \kappa 1^3
+ 285\, \kappa 2^2\, \kappa 3 + \left(270\, \kappa 2\, \kappa 3 + 10\, \kappa 3 + 30\, \kappa 5\right) \kappa 1^2 + 105\, \kappa 3\, \kappa 4 + \left(180\, \kappa 2^3
+ 15\, \kappa 2^2 + 210\, \kappa 2\, \kappa 4 + 150\, \kappa 3^2 + 5\, \kappa 4 + 15\, \kappa 6\right) \kappa 1 + \left(10\, \kappa 3 + 60\, \kappa 5\right) \kappa 2 + \kappa 5
+ 3\, \kappa 7\right) \mu\left(R0 + \alpha\right) \right) \Big]
$$